Εντολή repeat

Εισαγωγή

Έχουμε ήδη γνωρίσει την επαναληπτική εντολή while που εκτελεί μια ομάδα εντολών όσο μια συνθήκη είναι αληθής και την εντολή for που όπως είδαμε είναι πιο βολική όταν είναι καθορισμένος ο αριθμός των επαναλήψεων που θα εκτελεστούν οι εντολές.

Μια ακόμη επαναληπτική εντολή της Pascal έρχεται να μας βοηθήσει ώστε σε ορισμένες περιπτώσεις να διατυπώνουμε μια επανάληψη με πιο βολικό τρόπο, εκφράζοντας καλύτερα τον τρόπο με τον οποίο σκεφτόμαστε.

Τώρα θα γνωρίσουμε την εντολή repeat η οποία εκτελεί μια ομάδα εντολών μέχρι να ικανοποιηθεί κάποια συνθήκη.

Θα εξετάσουμε πάλι το πρόβλημα της εύρεσης του αθροίσματος: 

S=1+2+3+4+ . . .+100.

Ένας άλλος τρόπος διατύπωσης του αλγορίθμου για την λύση είναι ο εξής. Δώσε αρχική τιμή στο άθροισμα μηδέν (s=0) και επανέλαβε την εντολή s:=s+m θέτοντας αρχική τιμή στο m=1 και αυξάνοντας κάθε φορά το m κατά 1 μέχρι το m να γίνει μεγαλύτερο του 100.

Το ποιο κάτω πρόγραμμα Pascal λύνει το πρόβλημα χρησιμοποιώντας την εντολή repeat.

program sum_with_repeat;

var m,s:integer;

begin

  s:=0;
{αρχικές εργασίες}

  m:=1;

  repeat

    s:=s+m;
{κύρια επεξεργασία}

    m:=m+1;

  until m>100;

  writeln ('το άθροισμα =',s);
{τελικές εργασίες}

end.

Στο πρόγραμμα αφού δώσω αρχική τιμή στον αθροιστή s=0 και στη μεταβλητή που προστίθεται σε κάθε επανάληψη m=1 στη συνέχεια επαναλαμβάνονται οι εντολές s:=s+m; και m:=m+1; δηλαδή προσθέτω τον προσθετέο m και αυξάνω την τιμή του προσθετέου κατά 1 για το επόμενο βήμα, μέχρι ο προσθετέος να γίνει μεγαλύτερος του 100 (δηλ. 101).

Η συνθήκη m>100 ελέγχεται σε κάθε επανάληψη μετά την εκτέλεση των εσωτερικών εντολών και μόλις ικανοποιηθεί (πάρει τιμή TRUE) σημαίνει ότι η εντολή repeat ολοκληρώθηκε και συνεχίζουμε με την επόμενη εντολή που στο πρόγραμμα είναι η εμφάνιση του τελικού αθροίσματος.

Το σχήμα συλλογισμού της repeat ανταποκρίνεται στο εξής

επανέλαβε τις εντολές ... μέχρι να βρεθεί αληθής η <συνθήκη>.

Σε Pascal το σχήμα συλλογισμού είναι:

repeat 

   εντολή;

   εντολή;

     ... 

until   <συνθήκη>

που σημαίνει να εκτελούνται οι  εντολές μεταξύ των λέξεων (ειδικών συμβόλων) repeat, until μέχρι να γίνει αληθής η συνθήκη που ακολουθεί το until.

Σύνταξη:

Η σύνταξη της εντολής repeat είναι πολύ απλή και περιγράφεται από το εξής συντακτικό διάγραμμα.


Όπως βλέπουμε αποτελείται από τα ειδικά σύμβολα repeat και until μεταξύ των οποίων γράφουμε τις εσωτερικές εντολές που ξεχωρίζουν μεταξύ τους, όπως πάντα με το ειδικό σύμβολο ";". Μετά το until πρέπει να υπάρχει η συνθήκη ελέγχου της επανάληψης (μια λογική παράσταση).

Λειτουργία:

Ο τρόπος με τον οποίο εκτελείται η εντολή repeat φαίνεται στο παρακάτω λογικό διάγραμμα.

	
	

	
	

	

	
	λογικό διάγραμμα της  εντολής repeat
	
	λογικό διάγραμμα της εντολής repeat απλοποιημένο
	


Βλέπουμε ότι πρώτα εκτελούνται η εσωτερικές εντολές (αυτές που περιέχονται μεταξύ του repeat και του until) και μετά ελέγχεται η τιμή της Λ.Π.  που βρίσκεται μετά το until. Αν δεν είναι αληθής (δηλ. είναι FALSE) ξαναεκτελούνται οι εντολές και ελέγχεται και πάλι η τιμή της Λ.Π.   Αυτό επαναλαμβάνεται μέχρι η τιμή της Λ.Π.  να βρεθεί αληθής, οπότε η repeat τερματίζεται και συνεχίζουμε με την επόμενη εντολή.

Παρατηρούμε ότι η ουσιαστική διαφορά της repeat από την while είναι ότι ο έλεγχος της συνθήκης γίνεται μετά την εκτέλεση της εντολής σε αντίθεση με την while που γίνεται στην αρχή. Σαν αποτέλεσμα αυτού έχουμε ότι οι εντολές που περιέχονται στην repeat-until εκτελούνται τουλάχιστον μία φορά.

Κάθε εντολή repeat είναι ισοδύναμη με μία while ως εξής:

όπου s1,s2,. . .,sn εντολές και p μια λογική παράσταση.

	Repeat

  S1;S2;.....;Sn

until p;
	
	S1;S2;.....;Sn

while not p do

begin

S1;S2;....;Sn

end;
	


Υπάρχουν όμως εντολές while όπου δεν υπάρχει ισοδύναμη repeat αυτό συμβαίνει στις περιπτώσεις όπου οι εσωτερικές εντολές δεν πρέπει να εκτελεστούν καμία φορά.

Διάφορα παραδείγματα:

1) Το παρακάτω πρόγραμμα διαβάζει αριθμούς μέχρι το άθροισμά τους να γίνει μεγαλύτερο από 1000. Εμφανίζει το τελικό άθροισμα.

program test;

uses wincrt;

var

  number,s:integer;

begin

  s:=0;

  repeat

    write('δώσε αριθμό:');

    readln(number);

    s:=s+number;

  until s>1000;

  write('άθροισμα=', s);

end.

2) Να γραφεί πρόγραμμα που θα διαβάζει  ακέραιους θετικούς αριθμούς που θα δίνει ο χρήστης από το πληκτρολόγιο μέχρι να δοθεί ο αριθμός 0. Το πρόγραμμα να βρίσκει και να εμφανίζει τον μεγαλύτερο. Σε περίπτωση που ο χρήστης κάνει λάθος και δώσει αρνητικό αριθμό το πρόγραμμα να επιμένει μέχρι να πάρει θετικό αριθμό ή 0 για τέλος.

program megalyteros;

uses wincrt;

var

  number, max:integer;

begin

  writeln('βρίσκω τον μεγαλύτερο θετικό αριθμό ',

           'από όσους μου δώσεις ');

  writeln('δώσε 0 για να τελειώσει το πρόγραμμα ');

  max:=0;    { το 0 είναι μικρότερο από όλους τους }
                                 {πιθανούς αριθμούς}

  repeat

    repeat    {εξασφαλίζει ότι θα δοθεί θετικός ή 0}
      write('δώσε θετικό αριθμό (0=τέλος):');

      readln(number);

    until number>=0;

    if number>max then max:=number;
  until number=0;

  if max=0 then

     writeln('δεν δόθηκε κανένας θετικός αριθμός')

  else write('ο μεγαλύτερος είναι ο ',max);

end.

1>{διαβάζει ένα αριθμό εξασφαλίζοντας ότι είναι από 1 έως 100)

repeat
  write('Δώσε αριθμό(1-100):');

  readln(a);

until (a>=1) and (a<=100);

2>{περιμένει μέχρι να πατήσω ένα πλήκτρο(turbo Pascal μόνο)}

repeat until keypressed; 

Εδώ η εντολή μεταξύ των repeat και until είναι η κενή εντολή.

η keypressed είναι μια συνάρτηση της Turbo Pascal (στην wincrt) η οποία επιστρέφει την τιμή true όταν πατηθεί κάποιο πλήκτρο. Έχει την τιμή false πριν πατηθεί κάποιο πλήκτρο. 

3>{άθροισμα αριθμών μέχρι να δοθεί μηδέν}

repeat

  readln(A);

  s:=s+A;

until A=0;

4>{εξασφαλίζει ότι δεν θα διακοπεί το πρόγραμμα αν δοθεί μη αποδεκτή τιμή στο A (turbo Pascal μόνο)}

repeat

  {$I-} readln(A); {$I+}

until ioresult=0;

η  ioresult είναι συνάρτηση η οποία επιστρέφει τον αριθμό  0 (μηδέν) όταν η προηγούμενη εντολή εισόδου ήταν επιτυχής.

τα {$I-} και {$I+} είναι οδηγίες στον μεταγλωττιστή να απενεργοποιήσει ({$I-}) τον μηχανισμό διακοπής του προγράμματος σε περίπτωση λάθους και μετά να τον ενεργοποιήσει πάλι({$I+})  

Παρατηρήσεις

Παρατηρούμε ότι δεν χρειάζεται να βάζουμε begin, end για να ομαδοποιήσουμε τις εσωτερικές εντολές, αυτό το ρόλο παίζουν τα σύμβολα repeat και until

Παρ' όλα αυτά δεν είναι λάθος αν τα χρησιμοποιήσουμε.(γιατί;)

Τελειώνοντας ας σημειώσουμε ότι η εντολή repeat πρέπει να χρησιμοποιείται κυρίως στις περιπτώσεις που ο αριθμός των επαναλήψεων δεν είναι γνωστός εκ των προτέρων και στις οποίες είναι πιο βολική ή πιο φυσιολογική η εξέταση της λογικής παράστασης στο τέλος της ανακύκλωσης. 

Το πιο συνηθισμένο ίσως σημείο που χρησιμοποιείται η repeat είναι για να ελέγξουμε αν κάποια εντολή έχει διαβάσει κατάλληλη τιμή (βλ. παράδειγμα 1).

repeat





εντολή





until





Λογική παράσταση





;





εντολές





Λ.Π.





repeat





false





true





until





εντολές





Λ.Π





repeat





true








